Rise of Nations Scripting Language

Reference Manual
Background

The scripting language is derivative of C and JavaScript in most of its grammar and syntax. For performance, it is a compiled language. Scripts are currently used primarily to implement scenario triggers, to control some CtW scenarios and for some AI production controls. This document assumes the reader is already familiar with basic C/JavaScript syntax, and concentrates on the features particular to the scripting language.
This document is divided into two main portions,  the Essential Topics and the Advanced Topics.

The Essential Topics are pretty much all you need to get started with scripting, while the Advanced Topics provide some additional info on debugging and structure of the language itself.

Essential Topics
· Example of A Scenario Script
· Script Keywords & Operators
· Script Declaration
· Data & Variables
· Operators
· Boolean Evaluation
· Labels
· Triggers & run_once
· Calling RoN Script Functions
· Script Log & Errors
Advanced Topics
· Multiple Scripts per File
· Script Libraries & Include Files
· Debugging Scripts with Script Editor
· eval Command
· Data Types
· Auto Type-Casts
· Returns
· Parameter Types
· XML Text Extraction
Essential Topics

Example of a Scenario Script
First, a brief overview of how scripts are used and an example of a script:

The most common script used is a scenario script. When a scenario is made, you have the option of adding a script to go along with the Scenario. You can view or create the script associated with a particular scenario by clicking Edit Script. A script can be constructed entirely through the Script Editor. The associated script always has the same name as the scenario, but a .bhs extension. So myscenario.sce would have myscenario.bhs as the corresponding script.
The script is compiled while the game is loaded, and if there are any compiler errors, they will be displayed as soon as the game is loaded. 

If there are no errors, after a scenario is loaded, the script will run once every game, usually responding to game events in the form of triggers. (More on Triggers).
Here’s the example of a script from sample_script.bhs:

//******************** Sample_Script.bhs ********************//

//This is a sample script file

//The following is a skeleton scenario script file

scenario

{

labels {

  //Add any constant labels here in a comma separated list.

}

run_once {

  //Insert commands that should be executed only once when the script is loaded here.

}

//Main script body.

//Add triggers and other commands to be executed each game frame here.

//Use Edit->Insert Trigger Function to select from the list of available functions

//Use Edit->Insert Trigger Block to add a blank trigger block.

trigger sample_trigger (/*conditions under which the trigger will fire*/) {

  /*Insert commands that will execute when trigger condition is true and the trigger is enabled.

    Use trigger functions from (see list by selecting Edit->Insert Trigger Function) to control game events*/ 

}

The above script does nothing but is a template to get started on scenario scripts. (This is the template script that will be created when making a new script through Edit Script option in the Scenario Editor.)

Script Keywords & Operators

The language used to create such scripts is fairly similar to C and JavaScript. A lot of the BHS grammar is adapted from C, and many C keywords and operators can be used exactly in the script. The following is a brief overview of the differences.

Keywords & Operators equivalent to those in C:

if, else, for, while, do … while, switch, case, break, continue

+, -, *, /, %, =, +=, -=, *=, /=, %=, = =, !=, <, <=, >, >=, !, &&, ||, ?: (conditional),

++, -- (postfix and prefix), int (signed), float

Not Supported in Script:

pointers, arrays, structs, bitwise ops, preprocessor directives, externs

char, short, double, long, anything unsigned

Supported Differently or Partially Supported:

Enum – supported, use “labels” keyword instead. Can also be placed outside script blocks and included. (see advanced topics)

Include – can use to include “states” blocks from other files. Can NOT include scripts (see advanced topics)

Global Vars – data can be persistent within a certain script.

Reference Vars – reference parameters can be declared as ref (see data types)

Explicit Casts – even though data types can be cast automatically, explicit casts are available from and to each type.

Additional Features

run_once flow control, multiple script types, built-in string data type, types and declarations optional, automatic casts, ^, ^= (power op)

Like C statements, script statements must end with a semi-colon. Blocks are represented by { } braces, and C-statements that are usually accompanied by blocks (if, else, while, for, do, switch) may have blocks in the script as well. String literals are enclosed in double-quote “” marks. Character literals are treated as in C; they are integer values. All integer and float values are 4-byte values.

Subsequent sections will focus on the script-specific features.

Script Declaration

A script consists of a set of commands within a scripttype block:

scripttype {
<all script commands go here>

}

 where scripttype is either scenario, conquest or ai. You will almost always be working with a scenario script, but here is a brief description of each.

scenario – this is the kind of script used for custom made scenarios in the Scenario Editor, it contains all the events and triggers specific to that scenario.

conquest – this is a script type used internally for CtW scenarios. These scripts exist in the conquest directory and their usage is specified in conquestrules.xml.

ai – there are a couple ai scripts used internally, these are concerned primarily with production and best-build algorithms for the AI.

Data & Variables

Data in the script is comprised of integer, floating point and string values. Script variables can automatically store any of these values. The variables can be declared or “created” simply by assigning a value to them. Variables naming are the same as in C: variables must start with a letter, may contain alphanumeric chars, and can be of any length.

As soon as a value has been assigned to variable with an assignment operator, the variable can be used.  Any value can be assigned to any variable declared in this way. The following are all valid assignments:

first_name = “Play”

last_name = “Tester”;

age = 25;

id = 112358132134

if (last_name == “Tester”) {


id = “A1234B45678”;

}
Also, note that all identifiers (script, variable, trigger, label names, etc…) in are case insensitive!

Operators
The script has an additional ^ (power) and ^= operator. These operators can be used as follows:

cubed = a ^ 3;

three_halves = cubed;

three_halves ^= 0.5;

The order of operations for script operators is the same as in C, with the ^ operator having a priority just above *, / and %. Order of operations can be altered with parentheses.

All operators are valid for integer and floating point values. Strings values can be concatenated with + and +=. In general, a + or += call involving a string and a numeric value will result in the concatenation of the numeric value to the string. Other operators are not valid for strings and will cause compiler errors.

(For more info, see Data Types in the Advanced section)

Boolean Evaluation
While there is no specific boolean type in BHS, true and false are valid keywords and are equivalent to 1 and 0, respectively. However, since all script functions from the game follow a convention of returning -1,there is a slightly different implicit boolean evaluation.
if(x) is equivalent to if(x > 0) NOT if (x != 0)  (as in C)

Thus, if(-1) would result in false and the body of the if statement will not be evaluated. This difference is made to conveniently handle return values of -1, which usually imply error, and to differentiate them from a return value of 0, which often simply means false.

Labels

Labels are very analogous to an enum in C++. They allow for easily defining a set of constants.

If no value is specified, the label values increment by 1 for each subsequent label. If no starting value is specified, the first value of label is 1.
Eg.

labels {

 HUMAN,

 COMPUTER,

 BARBARIAN

}

defines a set of constants such that HUMAN is 1, COMPUTER is 2, and BARBARIAN is 3. (It is convention to use all caps for such constants, but it is not required.) Of course, the value of these can not be changed nor can the same name be used again.

Unlike C++ enums, strings and real numbers and constant expressions can also be used in labels, but not anthing else:

labels {

NAME_1 = “player”,   //OK
CITY_NAME = “comp”,  //OK
PI = 3.14159,        //OK

TWO_CUBED = (2 ^ 3), //OK

SQRT_5 = sqrt(5),    //BAD, this line will cause an error, no function calls

}

Triggers & run_once
Triggers are a special kind of if statement, denoted with the trigger keywords instead of an if in the language and allow for more complex decision trees within scenarios.

Essentially, a trigger is an if statement, that also has the property of being enabled/disabled. The commands within the trigger block are executed only when the condition is true (exactly like an if statement) AND the trigger is enabled.

All triggers start out enabled, but a typical scenario situation is to disable all triggers at the beginning (with disable_all_triggers) and then enable them one by one (with the enable_trigger function) as the scenario unfolds.

Once a trigger has “fired” (the commands within the trigger block have been executed), it becomes disabled, but can again be enabled with the enable_trigger call. Similarly, to disable a trigger before it has even executed, use disable_trigger.

A trigger can be emulated with an if statement and an extra variable, but the trigger is used for efficiency and to reduce the number of variables. The general rule is, if you need commands that are executed dependent on a particular state of a scenario, use a trigger, if you’re testing a global state of the game that applies no matter what, use an if statement. Here is a rough if statement equivalent for the way trigger statements work.

	Using triggers
	Using ifs

	trigger some_name (condition) {

Some commands here...

}
	if (some_name_var_enabled && condition) {

Some commands here...

}

	enable_trigger(some_name)
	some_name_var = true

	disable_trigger(some_name)
	some_name_var = false

	enable_all_triggers
	No equivalent! Must go through all the boolean vars and set them to true

	disable_all_triggers
	No equivalent! Must go through all the boolean vars and set them to false


Run_Once
A run_once statement is a special kind of trigger that is guaranteed to run only once. It is immune to disable_all_triggers and enable_all_triggers. Everything withn a run_once { … } block will run exactly once, on the first frame of the game. (A disable_all_triggers call is often placed with a run_once block to turn off all triggers at the beginning of the game and turn them on later elsewhere)

Calling RoN Script Functions

Of course, the purpose of the scripting language is to provide a powerful and flexible way to control and customize the game. The key to this is a library of almost 600 native game functions specifically designed to be accessible by scripts.

A complete listing of these functions may be found by selecting Insert Trigger Function… from the Edit menu in the Script Editor and a detailed documentation on these functions is provided in the RoN Script Function Reference.
These functions form the crux of the script’s functionality, and a function call is the same as function call in C, consisting of the function name, parentheses and, depending on the function, a list of values for function parameters.

The int, float and string parameters listed in the documentation specify the type of  script value expected for each parameter. Note that If a different value is specified, it will automatically be converted to the correct type, so that you need not worry about data types. For example, a call: 

popup_dialog(“Test Dialog”,  1 + 3 + 4 +5);

is equivalent to:

popup_dialog(“Test Dialog”,  “13”);
(See Data Types in the Advanced section for more detailed discussion of data types)

Script Log & Errors

Compiler Errors

When a script fails to compile, compiler error messages are added to the Script Log and a brief message stating that there are errors is displayed, usually just as the game is finished loading.

The default key for displaying the Script Log is Ctrl + Alt + X.
All script related errors and messages will be added to the log. The text in the log can be copied or saved to a file for future reference. 

Runtime Errors & Warnings
The script is designed to be flexible enough to avoid run time errors, so these errors should be much more rare than compiler errors.  However, in complex scripts, such errors do sometimes occur (a good example of one would be a divide by 0). The game handles run time errors much like compiler errors; a message that an error has occurred will appear in the message list at the top left, and the error details will be appended to the Script Log.
If you’re running a script in the Script Editor, the errors will actually appear in a special error window at the bottom of the screen. They will not go to the main Script Log

Advanced Topics
Multiple Scripts Per File

The simplest way to define a script is with scenario, conquest or ai keyword.

However, a script may also have a name and take certain parameters. It can then be called much like a RoN native script function.

When a script name is omitted, the script actually assumes the name of the file

some_script_file.bhs:
equivalent to:
some_script_file.bhs:

scenario
scenario some_script_file
{
{

…
…

}
}

Since the game expects a scenario script to have the same name as the scenario file and be located in the .bhs file corresponding to the .sce file, it is better to declare the scenario script with a simple scenario {…} declaration.

However, to include several scripts in a file, it is necessary to provide with all but one of them with a name. Any two scripts in a file can call each other regardless of their order. For example, here is a set of two scripts that compute and display Fibonacci numbers:

scenario fibonacci

{

  static old_term = 0;

  //display the first 10 terms of Fibonacci each time the script is called. 

  for (term = old_term; term < (old_term + 12); term++) {

    result = fib(term);

    str = "Fibonacci Term " + term + ": " + result;

    print_msg (str);

  }

  old_term = term;

}

scenario fib (term)

{

  p1 = (term > 1 ? fib(term - 1) : 0);

  p2 = (term > 2 ? fib(term - 2) : 1);

  return (p1 + p2);

}
(To execute, select paste and save file into Script Editor, and then execute the fibonacci script.)

Script Libraries & Include Files
It is possible to create libraries of scripts through the use of include files.

To include a file, simply specify it with the include keyword, eg:

include “tut_lib.bhs”
The compiler will first look in the same directory as the original script, and then in all directories specified by the ScriptIncludePath entry in rise.ini. (The paths can be entered as a ; separated list)

If the included file can not be found, the compiler will display an error and halt.

Once included, a script can be called exactly the same as if it was in the same file, using the standard function call syntax. For examples of reusable script libraries, see tut_lib.bhs and aibestbuildlibrary.bhs, as well as the various tutorial and ai scripts.

Debugging Scripts with Script Editor
The Script Editor, in addition to editing and compiling scripts, provides the ability to debug any script as its running in a single player game (debugging during multiplayer is disallowed as it can lead to unpredictable results and disconnection). This section assumes the reader is familiar with debugging terminology (stepping, breakpoints, watches, etc…)

The most common way to debug a script is follows:

1. Open the Script Editor in game, the default keystroke for this is CTRL + ALT + Z.

2. From the Loaded Scripts drop down, select the script you wish to debug.

3. Set a breakpoint at a particular line in the script.

4. Click the X at the top right to return to the game.

5. When the script reaches the breakpoint, you will see a message asking you if wish to debug or skip this breakpoint.

6. If you click debug, the Script Editor will reappear and you will see additional buttons, with various Step commands and watch windows.

7. Continue debugging as needed, viewing and changing data through watch windows, etc…

8. When done debugging, simply hit X on the Script Editor and the game will resume.

Some Notes on Debugging:

· It is also possibly to directly step into or execute a script from the editor. The result is the same as waiting for the game to call the script and let it hit the breakpoint. The only difference is if parameters are needed for the script, the game will provide them, otherwise you must enter them manually.

· It is of course possible to open any script and execute/debug it from the script editor.

· If you need to debug a script on the very first frame of a game, you may open the Script Editor from the Tools menu, open the script and set the breakpoint. The breakpoints will be preserved as you enter the game OR.

· To set a hard-coded breakpoint, use the debug_break() script function. It is equivalent to setting a breakpoint manually, except that it can not be removed without editing the script.

· It is possible to edit the script on the fly while and recompile it. This is a good way to test and debug scenario scripts without restarting each time. The changes will not take effect until the next frame, or the next time the script is restarted. Static variables and trigger states will be preserved when a script is edited. Remember to save and compile the script for the changes to take effect.

Eval Command
Another useful debugging feature is the console Eval command.

The command allows one to call a RoN script function directly, without using a script.

It was created to test the script functions, but it is also a good way to check and see exactly what a particular function does.

To use the command, follow these steps:

1. To gain access to the console, you must first enable cheats (just type cheat keys on at the chat prompt).

2. Display the console by pressing the ~ key.

3. Simply type eval expression and press Enter. Any output from the function will be displayed on the subsequent line of the console.
This will evaluate the expression and essentially allows for on the fly execution of a line of script code. There is no limit on the complexity of the expression but it may only consist of function calls and constants. It is not possible to access or declare any script variables, triggers, etc… from eval.
Eval is most useful for simply calling one of the script functions directly to test out exactly what it does
Some examples of valid eval usage:

eval sqrt(2) + sin(.3)

eval research_tech("Gunpowder Age")

eval zoom_in(20, 20)

eval 9 + 10

eval (2^(1+3)) - (6.4 / 3)

An invalid eval call will result in an error message printed in the console instead of the return or result of the expression.
[Top]
Data Types

The scripting language has three basic data types, int, real, and string
In addition, to support, typeless variables, there is an anytype type which is the default type if no explicit type is specified. Any value can be assigned to an anytype variable.

int and real declarations are initialized to 0, string is initialized to empty string. anytype values are not initialized.

Local vars have a default “local” scope. There is also a “ref” scope, which can only be used for script parameter declarations, and a “static” scope, which can be used to make data persistent between multiple script executions. The static type is very useful for data that must persist through several script executions, such as for a scenario script. Generally, most variables in a scenario script will be declared static so that they persist throughout the scenario, or at least until a subsequent script execution.
The following declarations are all equivalent:

	Int variable
	Float Variable
	String variable
	“Anytype” Variable

	int a;
	real b;
	string s;
	v = 0

	int a = 0;
	real b = 0;
	string s = “”
	anytype v = 0

	local int a;
	local real b;
	local string s;
	local v = 0

	local int a = 0
	local real b = 0
	local string s = “”;
	local anytype v = 0


Disclaimer on anytype declaration

“anytype” or undeclared values, can be used to simplify scripting. As type checking is ignored, an “anytype” var can be an int, a float, a string, or be undefined. Each type can be cast to any other type when assigned to a typed variable (an int, a float, or a string), in certain cases. To get additional type safety when necessary, script system functions “is_int”, “is_string”, “is_float” can be used to verify the current value of an anytype var. Declarations such as antype x are not associated with any type, until some value is assigned to x.

Auto Type-Casts

If types are used and type conflicts occur, the data is converted or “cast” to the appropriate type automatically. The following is a list of conversations with some sample data.
	to string
	to real
	to int

	(int -> string) 57 -> “57”
	(int -> real)  57
	(real -> int) 57.4 -> int

	(real-> string) 57.4 -> “57.4”
	(string -> real) “234.5” -> 234.5
(string -> real) “abc” -> 0.0
	(string -> int) “234.5” -> 234
(string -> real) “abc” -> 0


(note: conversions to and from real numbers may not be exact do to floating point rounding inaccuracies)
Although this chart is provided as reference, actual data type conversions in game are usually unnecessary and are very rare. Nevertheless, checking for proper data may help reduce problems and avoid bugs.

Returns

A script can also be declared with a specific return type. When no return type is specified, an anytype return type is assumed. The script has the option to return a value. If no return statement is found, the script always returns the default of its return type. If the return type is anytype, the script returns 0 by default. If an incorrect return value is returned, the value is automatically converted to the proper type. The game does not check for return values from scenario scripts and the return value is generally useful for script library or subroutine scripts.

Parameters Types
Scripts can optionally be declared with parameter list. The parameter are declared as in C, data type followed by variable name. Parameters can be of any data type, including anytype. There is no limit on the number of parameters that a script can declare, however, all parameters must be provided in a call. There are no optional parameters. Generally, parameters are useful only for scripts that serve as functions or subroutines for other scripts. They are used for some AI scripts that receive data from the game. Scenario scripts do not and should not take any parameters.
Reference Parameters

Parameters can be passed by reference and will allow the script to modify the original data. However, at this time, the parameters must be declared with a specific type. A ref anytype combination is not allowed, as anytype params can not be passed by reference. The 3 valid reference param types are:
scenario foo(ref int a, ref float b, ref string c)
Invalid ref param types:
scenario bad(ref a, ref anytype b) //bad
Reference parameters may be useful for library or subroutine scripts. They are used in some of the AI script library routines.
Variable Sized Parameter-Lists
A few of the RoN native script functions may take a variable number of parameters. These are mostly system functions and are unlikely to be used. Such functions are listed in the function docs with a … as the last parameter. These functions are called like all other functions, except that the compiler does not require a limit on the amount of parameters and will accept above the minimum required params. Ex:

   parse(“Some $REPLACEME0 with $SOMETEXT1”, “thing”, “something else”);

   parse(“Your name is $BLAH0”, what_my_name_is);

(Parse is used to replace all words that start with $ and end with a number n, with the nth parameter following after the first. n is 0 based)

XML String Extraction
The compiler allows for certain strings to be marked and extracted from an XML file instead of directly from a script. This system was used to support localization for various scripts that shipped with RoN, but it might be of some use to scenario designers who wish to create multilingual versions of their scenarios.

If a string is surrounded by $S(), the compiler will look for that string in an XML file of the same name as the script but with a .xml extension. If such a file does not exist and/or if the string is not found in the file, the file will be created and the string will be added to a list of strings. The xml entry also contains a hash of the original string, such that the text in the XML can be changed (presumably to another language, or perhaps to something totally different?). Then, when the compiler encounters the string in $S(), it will use the text from the xml instead of the text in quotes.

ex. compiling the following script in some_script.bhs:

scenario {

print_msg($S(“Some Text”));

}

will result in an XML file some_script.xml that contains a hash key and the text “Some Text”. If “Some Text” in the XML file is changed, the next time the script is compiled, the text from the XML file is used in place of the “Some Text” in quotes above.
The .xml file is generated only if there is at least one $S string in the file.

Although this was intended for internal localization use, it might be useful to some people for further customization and flexibility.

